
Nettoyer avec DISM:

Executer le nettoyeur de disque:

 

Script de nettoyage:

Nettoyer Windows

dism /online /cleanup-image /startcomponentcleanup /resetbase

cleanmgr /sageset:50
cleanmgr /sagerun:50

@echo off
title Nettoyage Windows avant image
color 0A

echo ========================================
echo   Nettoyage Windows avant capture image
echo ========================================
echo.

:: 1. Supprimer fichiers temporaires
echo Suppression fichiers temporaires...
del /f /s /q "%TEMP%\*" >nul 2>&1
del /f /s /q "C:\Windows\Temp\*" >nul 2>&1

:: 2. Arrêter Windows Update et BITS
echo Arret des services Windows Update...
net stop wuauserv >nul 2>&1
net stop bits >nul 2>&1

:: 3. Purger SoftwareDistribution
echo Purge du cache Windows Update...
rd /s /q "C:\Windows\SoftwareDistribution"



 

:: 4. Relancer les services
net start wuauserv >nul 2>&1
net start bits >nul 2>&1

:: 5. Nettoyer WinSxS avec DISM
echo Nettoyage WinSxS...
Dism.exe /Online /Cleanup-Image /StartComponentCleanup /ResetBase

:: 6. Désactiver la mise en veille prolongée
echo Desactivation mise en veille prolongée...
powercfg -h off

:: 7. Supprimer les points de restauration
echo Suppression des points de restauration...
vssadmin delete shadows /all /quiet

:: 8. Vider la corbeille
echo Vidage de la corbeille...
rd /s /q C:\$Recycle.Bin

:: 9. Purge des journaux d'evenements
echo Nettoyage des journaux Windows...
for /F "tokens=*" %%G in ('wevtutil el') DO wevtutil cl "%%G"

echo.
echo === Nettoyage termine ===
echo.

:: 10. Proposer de lancer Sysprep
set /p sysprep="Voulez-vous lancer Sysprep maintenant ? (O/N) : "
if /I "%sysprep%"=="O" (
    echo Lancement de Sysprep...
    start /wait C:\Windows\System32\Sysprep\Sysprep.exe /generalize /oobe /shutdown
)

pause

